

SCIENCE & VIE

MENSUEL N°884


MAI 1991

DROGUE: LE PLAISIR QUI TUE (6)



AWACS : LE MAITRE DU CIEL

M2578 - 884 - 20,00 F



145 FB - 6 FS - \$can 3.75 - 550 Ptas - 550 Esc - 28 Dh - 2.750 Dt - 4800 L - USA NYC \$ 4.25 - RCI: 1370 CFA - 940 CFP

Jouons au micro-whist

INFORMATIQUE AMUSANTE

Le programme de jeu que nous vous proposons de réaliser ce mois-ci s'inspire du jeu de whist. En fait vous aurez l'ordinateur comme partenaire. Bien que gérant la distribution des cartes, précisons tout de suite qu'il ne connaîtra absolument pas celles que vous aurez en main. Mais nous avons tout de même doté le programme d'une routine de détection de tricherie. Si, au cours d'une partie, vous trichez, l'ordinateur se fâchera très fort et, de dépit, "quittera la table de jeu".

Cela dit définissons les règles de ce jeu. De l'ensemble des cartes du jeu nous ne gardons que l'as, le roi, la dame, le valet et le "10" de chaque couleur. Notre jeu comporte donc au total 20 cartes. Au premier tour de jeu une seule carte est donnée à chaque joueur ; une troisième carte, retournée de manière à ce que sa couleur soit visible, définit l'atout. L'un des deux joueurs, en l'occurrence soit l'ordinateur soit vous, entame la partie en annonçant s'il pense faire le pli ou non. Notons qu'étant donné qu'il doit fatalement y avoir un perdant si, par exemple, c'est vous qui commencez, et que vous annoncez que vous ferez le pli, l'ordinateur devra annoncer la même chose. Inversement, si vous décidez que vous ne ferez pas le pli, l'ordinateur devra faire de même. Les annonces suivront évidemment la même règle si ce dernier entame le jeu. Pour résumer cette règle, on peut dire que, afin qu'il y ait un perdant à chaque tour, il faut que le total des plis annoncés n'égale pas le total des plis qui seront effectivement réalisés.

Une fois ces annonces faites, les joueurs poseront leur carte chacun leur tour. Si les deux cartes sont de la même couleur, la plus forte emportera le pli. Si l'un des joueurs "coupe", c'est-à-dire joue la couleur de l'atout, le pli lui reviendra. Enfin si le joueur adverse ne peut fournir la couleur demandée, il

perdra le pli. Le jeu, à ce niveau-là, est un simple jeu de bataille.

Le vainqueur du tour sera le joueur ayant rempli son contrat, c'est-à-dire ayant remporté le nombre de pli qu'il a annoncé en début de partie. C'est le joueur gagnant le tour qui relancera le tour d'après. Les tours suivants s'effectueront de la même manière mais avec deux cartes puis trois, etc. jusqu'à 10 par partenaire.

Ce dernier tour, puisque l'ensemble des cartes du jeu est distribué aux participants, se jouera sans atout.

Précisons, pour terminer, que les joueurs sont tenus de fournir la couleur demandée. Notamment il ne sera possible de couper que si on ne possède plus de carte de la couleur proposée par l'adversaire ; la coupe est obligatoire, il est impossible de se défausser.

Ces quelques règles fixées, passons à l'écriture de notre programme. Nous commencerons par réserver trois tableaux de mémorisation. Le premier d'entre eux constituera le jeu de cartes d'origine, les deux autres le jeu des participants. Ensuite seront redéfinis les symboles graphiques nécessaires à la représentation des cartes. Vient ensuite une routine de remplissage des tableaux des joueurs. Celle-ci

```

10 MODE 2: CLEAR
20 REM *****
30 REM *
40 REM * CREATION TABLEAUX DE MEMORISATION. *
50 REM *
60 REM *****
70 DIM C(5,4)
80 DIM O(5,4)
90 DIM J(5,4)
100 LET SCJ=0: LET SCO=0
110 REM *****
120 REM *
130 REM * REDEFINITION DES SYMBOLES GRAPHIQUES *
140 REM *
150 REM *****
160 SYMBOL AFTER 127
170 SYMBOL 128,255,0,0,0,0,0,0,0
180 SYMBOL 129,0,0,0,0,0,0,0,255
190 SYMBOL 130,192,192,192,192,192,192,192,192
200 SYMBOL 131,3,3,3,3,3,3,3,3
210 SYMBOL 132,3,28,112,192,192,192,192,192
220 SYMBOL 133,192,56,14,3,3,3,3,3
230 SYMBOL 134,192,192,192,192,192,112,28,3
240 SYMBOL 135,3,3,3,3,3,14,56,192
250 SYMBOL 136,1,7,47,119,255,113,33,15
260 SYMBOL 137,128,224,244,238,255,142,132,240
270 SYMBOL 138,1,3,31,255,31,3,1,0
280 SYMBOL 139,128,192,248,255,248,192,128,0
290 SYMBOL 140,60,254,255,127,31,7,3,1
300 SYMBOL 141,60,127,255,254,248,224,192,128
310 SYMBOL 142,1,3,7,31,127,253,121,3
320 SYMBOL 143,128,192,224,248,254,191,158,192
330 SYMBOL 144,24,24,24,24,255,126,60,24
340 SYMBOL 145,16,48,112,255,255,112,48,16
350 REM *****
360 REM *
370 REM * INITIALISATION DES TABLEAUX. *
380 REM *
390 REM *****
400 LET TOUR=1: LET CTO=0: LET CTJ=0
410 FOR X=1 TO 5
420 FOR Y=1 TO 4
430 LET C(X,Y)=1
440 LET O(X,Y)=0
450 LET J(X,Y)=0
460 NEXT Y
470 NEXT X
480 REM *****
490 REM *
500 REM * TIRAGE ALEATOIRE DES CARTES. *
510 REM *
520 REM *****
530 FOR D=1 TO 2
540 FOR I=1 TO TOUR
550 LET XA=INT(RND*5)+1
560 LET YA=INT(RND*4)+1

```

```

570 IF C(XA,YA)=0 THEN GOTO 550
580 LET C(XA,YA)=0
590 IF D=1 THEN LET O(XA,YA)=1
600 IF D=2 THEN LET J(XA,YA)=1
610 NEXT I
620 NEXT D
630 LET ATOU=0: LET FXJ=1: LET FYJ=1: LET JENG=1
640 GOSUB 1280
650 REM *****
660 REM *
670 REM * DISTRIBUTION DES CARTES. *
680 REM *
690 REM *****
700 FOR I=1 TO 5
710 LOCATE 75,I: PRINT " "
720 NEXT I
730 IF TOUR=10 THEN GOTO 1010
740 LET XA=INT(RND*5)+1
750 LET YA=INT(RND*4)+1
760 IF C(XA,YA)=0 THEN GOTO 740
770 LET ATOU=YA
780 REM *****
790 REM *
800 REM * AFFICHAGE DE LA CARTE ATOU. *
810 REM *
820 REM *****
830 LOCATE 75,1: PRINT CHR$(132);CHR$(128);CHR$(128);CHR$(133)
840 LOCATE 75,2: PRINT CHR$(130);" ";CHR$(131)
850 LOCATE 75,3: PRINT CHR$(130);" ";CHR$(131)
860 LOCATE 75,4: PRINT CHR$(134);CHR$(129);CHR$(129);CHR$(135)
870 IF YA=1 THEN LOCATE 76,3: PRINT CHR$(136);CHR$(137)
880 IF YA=2 THEN LOCATE 76,3: PRINT CHR$(138);CHR$(139)
890 IF YA=3 THEN LOCATE 76,3: PRINT CHR$(140);CHR$(141)
900 IF YA=4 THEN LOCATE 76,3: PRINT CHR$(142);CHR$(143)
910 IF XA=1 THEN LOCATE 76,2: PRINT "AS"
920 IF XA=2 THEN LOCATE 76,2: PRINT "R"
930 IF XA=3 THEN LOCATE 76,2: PRINT "D"
940 IF XA=4 THEN LOCATE 76,2: PRINT "V"
950 IF XA=5 THEN LOCATE 76,2: PRINT "10"
960 REM *****
970 REM *
980 REM * DEBUT DE PARTIE. *
990 REM *
1000 REM *****
1010 LET PJ=0: LET PO=0: LET COUP=0: LET PJF=0: LET POF=0
1020 LET PPJ=0: LET PPO=0: LET POF=0
1030 GOSUB 1590
1040 IF JENG=1 THEN GOSUB 1830: LET CD=YJ
1050 IF JENG=0 THEN GOSUB 2370: LET CD=Y0

```

sera évidemment liée au numéro du tour en cours afin de distribuer le nombre de cartes adéquat. Il sera contrôlé pour chaque nouvelle carte distribuée, à partir d'un tirage aléatoire dans le tableau du jeu de carte, qu'elle n'a pas déjà été donnée, cas évidemment impossible dans la réalité.

Cette distribution terminée, les cartes du joueur seront présentées à l'écran puis, s'il y a lieu, l'atout

sera tiré et également affiché. La partie pourra dès lors débiter. A son début la machine demandera si vous souhaitez commencer ou non. Un test sur INKEY\$ aiguillera alors le programme en fonction de la réponse fournie. Suivant le cas deux sous-routines différentes seront utilisées.

La première sous-routine concerne le choix de la carte par le joueur et l'annonce de son contrat.

En ce qui concerne ce dernier il précisera s'il pense effectuer le pli ou non. Puis les flèches de déplacement du curseur permettront de "pointer" la carte désirée d'après sa position sur la table en ligne et colonne. Le programme vérifiera alors que la carte visée existe bien. Son choix sera validé par la touche COPY, placée au centre des flèches de déplacement curseur. Cette opération effectuée, le programme

```

1060 GOSUB 2180
1070 IF JENG=0 THEN GOSUB 1830:GOSUB 2870
1080 IF JENG=1 THEN GOSUB 2370
1090 GOSUB 2180
1100 GOSUB 2640
1110 LET COUP=COUP+1:IF COUP<TOUR THEN GOTO 1040
1120 IF PPJ>=PJ THEN LET SCJ=SCJ+1
1130 IF PPO>=PO THEN LET SCO=SCO+1
1140 LET TOUR=TOUR+1
1150 LOCATE 60,6:PRINT "CONTRATS REMPLIS."
1160 LOCATE 60,8:PRINT "VOUS:":SCJ
1170 LOCATE 60,10:PRINT "MOI:":SCO
1180 LOCATE 1,3:PRINT "
1190 IF TOUR<11 THEN GOTO 410
1200 LOCATE 1,1:PRINT " UNE AUTRE PARTIE TQPER UNE
TOUCHE SVP."
1210 IF INKEY$="" THEN GOTO 1210
1220 GOTO 10
1230 REM *****
1240 REM *
1250 REM * SOUS ROUTINE D'AFFICHAGE DES CARTES. *
1260 REM *
1270 REM *****
1280 FOR V=1 TO 4
1290 FOR H=1 TO 5
1300 LET HF=((H-1)*5)+1:LET VF=((V-1)*5)+5
1310 IF J(H,V)=0 THEN GOTO 1360
1320 LOCATE HF,VF:PRINT CHR$(132);CHR$(128);CHR$(1
28);CHR$(133)
1330 LOCATE HF,VF+1:PRINT CHR$(130);" ";CHR$(1
31)
1340 LOCATE HF,VF+2:PRINT CHR$(130);" ";CHR$(1
31)
1350 LOCATE HF,VF+3:PRINT CHR$(134);CHR$(129);CHR$
(129);CHR$(135)
1360 NEXT H
1370 NEXT V
1380 FOR H=1 TO 5
1390 LET HF=((H-1)*5)+2
1400 IF J(H,1)=1 THEN LOCATE HF,7:PRINT CHR$(136);
CHR$(137)
1410 IF J(H,2)=1 THEN LOCATE HF,12:PRINT CHR$(138)
;CHR$(139)
1420 IF J(H,3)=1 THEN LOCATE HF,17:PRINT CHR$(140)
;CHR$(141)
1430 IF J(H,4)=1 THEN LOCATE HF,22:PRINT CHR$(142)
;CHR$(143)
1440 NEXT H
1450 FOR V=1 TO 4
1460 LET VF=((V-1)*5)+6
1470 IF J(1,V)=1 THEN LOCATE 2,VF:PRINT "AS"
1480 IF J(2,V)=1 THEN LOCATE 7,VF:PRINT "R"
1490 IF J(3,V)=1 THEN LOCATE 12,VF:PRINT "D"
1500 IF J(4,V)=1 THEN LOCATE 17,VF:PRINT "V"
1510 IF J(5,V)=1 THEN LOCATE 22,VF:PRINT "10"
1520 NEXT V
1530 RETURN
1540 REM *****
1550 REM *
1560 REM * SOUS ROUTINE D'ANNONCE DES CONTRATS. *
1570 REM *
1580 REM *****
1590 LOCATE 1,1:PRINT "COMBIEN DE PLIS ANNONCEZ VO
US";
1600 INPUT PJ
1610 IF PJ<0 OR PJ>TOUR OR (PJ+PO=TOUR AND JENG=0)
THEN GOTO 1600
1620 LET PO=0
1630 FOR I=1 TO 1+INT(TOUR/4)
1640 IF O(I,1)=1 THEN LET PO=PO+1
1650 IF O(I,2)=1 THEN LET PO=PO+1
1660 IF O(I,3)=1 THEN LET PO=PO+1
1670 IF O(I,4)=1 THEN LET PO=PO+1
1680 NEXT I
1690 IF ATOU=0 THEN GOTO 1740
1700 IF O(3,ATOU)=1 AND TOUR<9 THEN LET PO=PO+1
1710 IF O(4,ATOU)=1 AND TOUR<8 THEN LET PO=PO+1
1720 IF O(5,ATOU)=1 AND TOUR<7 THEN LET PO=PO+1
1730 IF TOUR =1 THEN LET PO=1
1740 IF PO+PJ=TOUR THEN LET PO=PO+1
1750 IF PO>TOUR THEN LET PO=PO-2
1760 LOCATE 1,3:PRINT "J'ANNONCE";PO;"PLIS."
1770 RETURN
1780 REM *****
1790 REM *
1800 REM * SOUS ROUTINE DEPLACEMENT FLECHES. *
1810 REM *
1820 REM *****
1830 LET XFL=1:LET YFL=1:LET XAFFL=2:LET YAFFL=6
1840 LOCATE 60,14:PRINT "PLIS FAITS:"
1850 LOCATE 60,16:PRINT "VOUS:":PPJ
1860 LOCATE 60,18:PRINT "MOI:":PPO
1870 LOCATE XAFFL,4:PRINT "
1880 LOCATE 26,YAFFL:PRINT "
1890 IF INKEY(1)=0 AND XFL<5 THEN LET XFL=XFL+1
1900 IF INKEY(8)=0 AND XFL>1 THEN LET XFL=XFL-1
1910 IF INKEY(2)=0 AND YFL<4 THEN LET YFL=YFL+1
1920 IF INKEY(0)=0 AND YFL>1 THEN LET YFL=YFL-1
1930 LET XAFFL=2+(XFL-1)*5:LET YAFFL=6+(YFL-1)*5
1940 LOCATE XAFFL,4:PRINT CHR$(144)
1950 LOCATE 26,YAFFL:PRINT CHR$(145)
1960 IF INKEY$="" THEN GOTO 1960
1970 IF INKEY(9)=0 THEN GOTO 1990
1980 GOTO 1870
1990 IF J(XFL,YFL)=1 THEN GOTO 2030
2000 PRINT CHR$(7)
2010 LOCATE 1,25:PRINT "* CHOIX IMPOSSIBLE *"
2020 GOTO 1870
2030 LET J(XFL,YFL)=0:LET XJ=XFL:LET YJ=YFL
2040 LET HF=((XFL-1)*5)+1:LET VF=((YFL-1)*5)+5
2050 LOCATE XAFFL-1,YAFFL-1:PRINT " "

```

l'effacera du tableau du joueur. Elle s'affichera aussitôt sur le "tapis de jeu" et la carte jouée par la machine, une fois son choix effectué, apparaîtra au-dessus.

La seconde concerne le choix de la machine. Son annonce sera fonction de celle du joueur s'il joue en second ou de ses estimations s'il doit prendre la décision du jeu. Tout comme le joueur, l'ordinateur choisira sa carte en fonction de la

couleur demandée, de son annonce et, éventuellement, suivant ses "espoirs" s'il doit jouer le premier, et la posera sur la table.

Une fois les deux cartes sur la table, une troisième sous-routine déterminera lequel des deux joueurs remporte le pli et créditera le nombre de contrats remporté par l'un des participants en fonction des annonces.

Le jeu rebouclera ainsi 10 fois de

suite afin d'exécuter une partie complète.

La frappe de ce programme ne doit pas poser de problème particulier. Bien qu'il soit étudié sur un Amstrad 464, son adaptation à d'autres machines est simple. Il suffira simplement pour cela de remplacer les instructions propres à l'Amstrad, MODE, SYMBOL AFTER, etc. par leurs équivalentes.

Henri-Pierre Penel

```

2060 LOCATE XAFFL-1,YAFFL:PRINT "      "
2070 LOCATE XAFFL-1,YAFFL+1:PRINT "      "
2080 LOCATE XAFFL-1,YAFFL+2:PRINT "      "
2090 LET CTAP=15:LET COULT=YFL:LET FT=XFL
2100 LOCATE XAFFL,4:PRINT "      "
2110 LOCATE 26,YAFFL:PRINT "      "
2120 RETURN
2130 REM *****
2140 REM *
2150 REM * SOUS ROUTINE AFFICHAGE CARTE JOUEE. *
2160 REM *
2170 REM *****
2180 LOCATE 50,CTAP:PRINT CHR$(132);CHR$(128);CHR$(128);CHR$(133)
2190 LOCATE 50,CTAP+1:PRINT CHR$(130);" ";" ";CHR$(131)
2200 LOCATE 50,CTAP+2:PRINT CHR$(130);" ";" ";CHR$(131)
2210 LOCATE 50,CTAP+3:PRINT CHR$(134);CHR$(129);CHR$(129);CHR$(135)
2220 IF COULT=1 THEN LOCATE 51,CTAP+2:PRINT CHR$(138);CHR$(137)
2230 IF COULT=2 THEN LOCATE 51,CTAP+2:PRINT CHR$(138);CHR$(139)
2240 IF COULT=3 THEN LOCATE 51,CTAP+2:PRINT CHR$(140);CHR$(141)
2250 IF COULT=4 THEN LOCATE 51,CTAP+2:PRINT CHR$(142);CHR$(143)
2260 IF FT=1 THEN LOCATE 51,CTAP+1:PRINT "AS"
2270 IF FT=2 THEN LOCATE 51,CTAP+1:PRINT "R"
2280 IF FT=3 THEN LOCATE 51,CTAP+1:PRINT "D"
2290 IF FT=4 THEN LOCATE 51,CTAP+1:PRINT "V"
2300 IF FT=5 THEN LOCATE 51,CTAP+1:PRINT "10"
2310 RETURN
2320 REM *****
2330 REM *
2340 REM * SOUS ROUTINE DE JEU DE L'ORDINATEUR. *
2350 REM *
2360 REM *****
2370 LET XO=0
2380 FOR I=1 TO 5
2390 IF O(I,YFL)=1 AND I<XFL AND POF<PO THEN LET XO=I
2400 NEXT I
2410 IF XO<>0 THEN LET YO=YFL:GOTO 2550
2420 FOR I=5 TO 1 STEP -1
2430 IF O(I,YFL)=1 AND XO=0 THEN LET XO=I
2440 NEXT I
2450 IF XO<>0 THEN LET YO=YFL:GOTO 2550
2460 FOR I=5 TO 1 STEP -1
2470 IF O(I,ATO)=1 THEN LET XO=I
2480 NEXT I
2490 IF XO<>0 THEN LET YO=ATO:GOTO 2550
2500 FOR H=5 TO 1 STEP -1
2510 FOR V=1 TO 4
2520 IF O(H,V)=1 THEN LET XO=H:LET YO=V
2530 NEXT V
2540 NEXT H
2550 LET O(XO,YO)=0
2560 LET XFL=XO:LET YFL=YO
2570 LET CTAP=10:LET COULT=YO:LET FT=XO
2580 RETURN
2590 REM *****
2600 REM *
2610 REM * SOUS ROUTINE DETERMINATION GAGNANT. *
2620 REM *
2630 REM *****
2640 IF YJ=ATO AND YO<>ATO THEN LET JENG=1:GOTO 2700
2650 IF YO=ATO AND YJ<>ATO THEN LET JENG=0:GOTO 2700
2660 IF YJ=CD AND YO<>CD THEN LET JENG=1:GOTO 2700
2670 IF YO=CD AND YJ<>CD THEN LET JENG=0:GOTO 2700
2680 IF XJ<XO THEN LET JENG=1
2690 IF XO<XJ THEN LET JENG=0
2700 IF JENG=0 THEN LET POF=POF+1:LOCATE 1,25:PRINT "LE PLI EST POUR MOI."
2710 IF JENG=1 THEN LOCATE 1,25:PRINT "VOUS GAGNEZ LE PLI ."
2720 LOCATE 30,25:PRINT "POUR CONTINUER TAPER UNE TOUCHE."
2730 IF INKEY$="" THEN GOTO 2730
2740 IF JENG=1 THEN LOCATE 1,25:PRINT "LE JEU EST A VOUS..."
2750 IF JENG=0 THEN LOCATE 1,25:PRINT "-- JE VOUS PROPOSE : "
2760 FOR I=10 TO 20
2770 LOCATE 50,I:PRINT "      "
2780 NEXT I
2790 IF JENG=1 THEN LET PPJ=PPJ+1
2800 IF JENG=0 THEN LET PPO=PPO+1
2810 RETURN
2820 REM *****
2830 REM *
2840 REM * SOUS ROUTINE ANTI TRICHE. *
2850 REM *
2860 REM *****
2870 IF YFL=YO THEN GOTO 2950
2880 FOR I=1 TO 5
2890 IF J(I,YO)=1 GOTO 2960
2900 NEXT I
2910 IF YFL=ATO THEN GOTO 2950
2920 FOR I=1 TO 5
2930 IF J(I,ATO)=1 THEN GOTO 2960
2940 NEXT I
2950 RETURN
2960 MODE 0:PRINT "VOUS TRICHEZ!"
2970 FOR T=1 TO 5000:NEXT T
2980 CALL 0

```